

- Pazcal Grammar

The following rules are depicting the grammar of Pazcal language in EBNF-form :

```
1  <module>          ::= ( <declaration> ) *
2  <declaration>     ::= <const_def> | <var_def> | <routine> | <program>
3  <const_def>       ::= "const" <type> <id> "=" <const_expr> ( "," <id> "=" <const_expr> ) * ";"
4  <var_def>         ::= <type> <var_init> ( "," <var_init> ) * ";"
5  <var_init>        ::= <id> [ "=" <expr> ] | <id> ( "[" <const_expr> "]" ) +
6  <routine_header>  ::= ( "PROC" | "FUNC" <type> ) <id> "(" [ <type> <formal> ( "," <type> <formal> ) * ] ")"
7  <formal>          ::= [ "&" ] <id> | <id> "[" [ <const_expr> "]" <id> ( "[" <const_expr> "]" ) *
8  <routine>         ::= <routine_header> ( ";" | <block> )
9  <program_header> ::= "PROGRAM" <id> "(" "(" ")"
10 <program>         ::= <program_header> <block>
11 <type>            ::= "int" | "bool" | "char" | "REAL"
12 <const_expr>     ::= <expr>
13 <expr>            ::= <int-const> | <float-const> | <char-const> | <string-literal> | "true" | "false"
14                  | "(" <expr> ")" | <l_value> | <call> | <unop> <expr> | <expr> <binop> <expr>
15 <l_value>         ::= <id> ( "[" <expr> "]" ) *
16 <unop>           ::= "+" | "-" | "!" | "not"
17 <binop>          ::= "+" | "-" | "*" | "/" | "%" | "MOD" | "==" | "!=" | "<" | ">" | "<=" | ">="
18                  | "&&" | "and" | "||" | "or"
19 <call>            ::= <id> "(" [ <expr> ( "," <expr> ) * ] ")"
20 <block>           ::= "{" ( <local_def> | <stmt> ) * "}"
21 <local_def>       ::= <const_def> | <var_def>
22 <stmt>            ::= ";" | <l_value> <assign> <expr> ";" | <l_value> ( "++" | "--" ) ";" | <call> ";"
23                  | "if" "(" <expr> ")" <stmt> [ "else" <stmt> ] | "while" "(" <expr> ")" <stmt>
24                  | "FOR" "(" <id> "," <range> ")" <stmt> | "do" <stmt> "while" "(" <expr> ")" ";"
25                  | "switch" "(" <expr> ")" "{" ( "case" <const_expr> ":" ) + <clause> ) *
26                  [ "default" ":" <clause> ] "}"
27                  | "break" ";" | "continue" ";" | "return" [ <expr> ] ";"
28                  | <write> "(" [ <format> ( "," <format> ) * ] ")" ";"
29 <assign>          ::= "=" | "+=" | "-=" | "*=" | "/=" | "%="
30 <range>           ::= <expr> ( "TO" | "DOWNTO" ) <expr> [ "STEP" <expr> ]
31 <clause>          ::= ( <stmt> ) * ( "break" ";" | "NEXT" ";" )
32 <write>           ::= "WRITE" | "WRITELN" | "WRITESP" | "WRITESPLN"
33 <format>          ::= <expr> | "FORM" "(" <expr> "," <expr> [ "," <expr> ] ")"
```